
Zusammenfassung 

 

Dieser Standard beschreibt die Definition von Meta-Attributen als Teil eines INTERLIS-

Datenmodells. Mit den Regeln, die in der vorliegenden Spezifikation definiert werden, kön-

nen Meta-Informationen in ein INTERLIS-Datenmodell integriert werden, ohne dass dafür die 

INTERLIS-Sprachspezifikation [1] angepasst werden muss. 

Die Syntaxregeln für Meta-Attribute werden in das Regelwerk von INTERLIS eingepasst. 

Insbesondere wird definiert, an welchen Stellen in der bestehenden INTERLIS-Spezifikation 

neu Meta-Attribute definiert werden können. 

1 Status des Dokuments 

Das vorliegende Dokument wurde vom Expertenausschuss genehmigt. Es hat für das defi-

nierte Einsatzgebiet im festgelegten Gültigkeitsbereich normative Kraft. 

2 Einleitung 

2.1 Überblick 

Meta-Attribute erlauben eine Ergänzung der INTERLIS-Modelle mit zusätzlichen Angaben, 

die in der aktuellen Spezifikation von INTERLIS [1] nicht vorgesehen sind. 

Im Dokument zum INTERLIS 2-Metamodell [2] steht dazu: 

«Einem Metaelement können kompositorisch Metaattribute zugeordnet sein (Klasse "Me-

taAttribut"). Metaattribute sind weder durch die Sprache INTERLIS noch durch das Metamo-

dell genauer definiert. Sie dienen insbesondere dazu, dass Informationen, die über 

INTERLIS hinausgehen, als Bestandteil der Modelldaten beigefügt werden können. Me-

taattribute haben einen Namen (Attribut "Name"), der innerhalb der Metaattribute desselben 

Metaelementes eindeutig sein muss, und den beschreibenden Wert (Attribut "Value").» 

Im vorliegenden Dokument wird eine Syntax beschrieben, welche erlaubt, solche Meta-

Attribute direkt als Teil des Modells formulieren zu können. Dies im Gegensatz zum Meta-

modell, wo definiert wird, wie Meta-Attribute als Daten definiert bzw. in XML transferiert wer-

den. 

 

2.2 Vorteile 

Die Definition von Meta-Attributen direkt im Datenmodell hat den Vorteil, dass Metainformati-

onen unmittelbar bei denjenigen INTERLIS-Konstrukten erfasst werden können, die sie be-

einflussen sollen. Die vorliegende Spezifikation ist so konzipiert, dass Meta-Attribute ohne 

Änderung der INTERLIS-Spezifikation [1] angewendet werden können. 

Im Unterschied zu normalem Zeilenkommentar, der ausschliesslich informativen Charakter 

hat, können Meta-Attribute gemäss den Syntaxregeln aus Abschnitt 4.2 von entsprechend 

befähigten Softwarewerkzeugen interpretiert und ggf. für die weitere Verarbeitung weiterge-

geben werden. 

 

3 Konzept 

Um die bestehende Sprachdefinition nicht ändern zu müssen, werden die Meta-Attribute in-

nerhalb von INTERLIS-Zeilenkommentaren formuliert. Ein INTERLIS-Zeilenkommentar ist 

gemäss [1] wie folgt definiert:  

«Ein Zeilenkommentar wird mit zwei Ausrufezeichen eröffnet, die unmittelbar aufeinander 

folgen. Der Zeilenkommentar wird durch das Zeilenende abgeschlossen.» 

Um Meta-Attribute von echtem Kommentar unterscheiden zu können, ist als drittes Zeichen 

zwingend das Zeichen «@», einzufügen. Das heisst, ein Zeilenkommentar, der Meta-Attribute 

enthält, beginnt mit «!!@». 

Zu welchem Meta-Element (gemäss INTERLIS 2-Metamodell) ein solches Meta-Attribut ge-

hört, ergibt sich aus der Position des Zeilenkommentars. Das Meta-Attribut bezieht sich auf 

das unmittelbar nachfolgende Sprachkonstrukt wie beispielsweise eine Modell-, Themen- 

oder Klassendefinition. 

 

4 Syntax 

4.1 Syntax-Definitionen 

Die Syntaxregeln sind entsprechend der Extended Backus-Naur-Form (EBNF) definiert, wie 

auch die übrigen INTERLIS-Sprachregeln (siehe [1], Abschnitt 2.1).  

Zusätzlich werden folgende Syntaxregeln eingeführt: 

~ A Ein Zeichen aus allen möglichen Zeichen aber verschieden von den durch  

A  definierten 

'A' .. 'D' Ein Zeichen aus dem Bereich von Zeichen: «A», «B», «C» und «D» 

'\t' Tabulatorzeichen  Unicode: <CHARACTER TABULATION> 

'\f' Seitenvorschubzeichen  Unicode: <FORM FEED (FF)> 

'\r' Wagenrücklaufzeichen  Unicode: <CARRIAGE RETURN (CR)> 

'\n' Zeilenvorschubzeichen  Unicode: <LINE FEED (LF)> 

 

4.2 Syntaxregeln für Meta-Attribute 

Ein Meta-Attribut besteht aus dem einleitenden Code «!!@» sowie einem Meta-Attribut-

Namen und einem zugewiesenen Meta-Attribut-Wert: 

MetaAttribute = '!!@' Name '=' Value { ';' [ Name '=' Value ] } . 

Ein Meta-Attribut-Wert wird entweder aus einem Namen oder einem freien String gebildet: 

Value = Metaattributename | String . 

Ein Metaattributename ist eine beliebige Zeichenfolge (max. 255 Zeichen Länge) ohne be-

stimmte Sonderzeichen: 

Metaattributename = (* ~( '\t' | '\f' | '\r' | '\n' | ' ' | '='  

   | ';' | ',' | '"' | '\' ) *) . 

Ein String ist eine beliebige Zeichenfolge inklusive «Escape»-Sequenzen; eingeschlossen in 

doppelte Anführungszeichen: 

String = '"' { Escape | ~( '"' | '\' ) } '"' . 

Eine Escape-Sequenz beginnt mit einem Backslash, gefolgt entweder von einem doppelten 

Anführungszeichen oder einem Backslash oder einer vierstelligen Hex-Unicode-

Zeichennummer: 

Escape = '\' ( '"' | '\'  

   | 'u' HexDigit HexDigit HexDigit HexDigit ) . 

Ein Unicode-Zeichen wird aus Hexadezimalzahlen gebildet. Diese bestehen aus Ziffern und 

den Buchstaben a/A bis f/F: 

HexDigit = '0' .. '9' | 'a' .. 'f' | 'A' .. 'F' . 

5 Positionierung innerhalb der INTERLIS-Sprache 

In den folgenden Ausschnitten aus Syntaxregeln aus dem INTERLIS-Referenzhandbuch [1] 

wird definiert, an welcher Stelle Meta-Attribute vorkommen dürfen. Meta-Attribute an anderen 

Stellen werden als reguläre Zeilenkommentare aufgefasst und deshalb ignoriert. 

Modell-Definition: 

ModelDef = { MetaAttribute } [ 'CONTRACTED' ] 

   [ 'TYPE' | 'REFSYSTEM' | 'SYMBOLOGY' ] 'MODEL' Model-Name … 

Topic-Definition: 

TopicDef = { MetaAttribute } [ 'VIEW' ] 'TOPIC' Topic-Name … 

Klassen-Definition: 

ClassDef = { MetaAttribute } 'CLASS' Class-Name … 

Struktur-Definition: 

StructureDef = { MetaAttribute } 'STRUCTURE' Structure-Name … 

Attribut-Definition: 

AttributeDef = { MetaAttribute } [ [ 'CONTINUOUS' ] 'SUBDIVISION' ] 

   Attribute-Name … 

Rollen-Definition in Assoziationen: 

RoleDef = { MetaAttribute } Role-Name … 

Wertebereichs-Definition: 

DomainDef = 'DOMAIN' { { MetaAttribute } Domain-Name … 

Aufzähltyp-Element: 

EnumElement = { MetaAttribute } EnumElement-Name … 

Linienform-Definition: 

LineFormTypeDef = 'LINE' 'FORM' { { MetaAttribute }  

   LineFormType-Name … 

Einheiten-Definition: 

UnitDef = 'UNIT' { { MetaAttribute } Unit-Name … 

Definition eines Metaobjekt-Behälters: 

MetaDataBasketDef = { MetaAttribute } ( 'SIGN' | 'REFSYSTEM' ) 

   'BASKET' Basket-Name Properties <FINAL> 

   [ 'EXTENDS' MetaDataBasketRef ] '~' TopicRef  

   { 'OBJECTS' 'OF' Class-Name ':'  

   { MetaAttribute } MetaObject-Name  

   { ',' { MetaAttribute } MetaObject-Name } } ';'. 

Parameter-Definition: 

ParameterDef = { MetaAttribute } Parameter-Name … 

Laufzeitparameter-Definition: 

RunTimeParameterDef = 'PARAMETER' { { MetaAttribute }  

   RunTimeParameter-Name … 

Definition einer Konsistenzbedingung: 

ConstraintDef = { MetaAttribute } ( MandatoryConstraint | … 

Funktions-Definition: 

FunctionDef = { MetaAttribute } 'FUNCTION' Function-Name … 

Definition einer Sicht (View): 

ViewDef = { MetaAttribute } 'VIEW' View-Name … 

Grafik-Definition: 

GraphicDef = { MetaAttribute } 'GRAPHIC' Graphic-Name … 

Meta-Attribute zu einer Topic-Definition werden in den Schemadaten, das heisst, den Me-

taobjekten gemäss Metamodell [2] dem «SubModel» zugeordnet (und nicht der «DataUnit»). 

Meta-Attribute zu einer Konsistenzbedingungs-Definition («ConstraintDef») sind als Sche-

madaten nicht möglich, weil Constraints gemäss Metamodell [2] keine OID haben bezie-

hungsweise kein «MetaElement» sind. 

6 Beispiel 

Folgendes Beispiel soll den Gebrauch von Meta-Attributen illustrieren. Dabei wird ange-

nommen, in einem INTERLIS-Datenmodell soll der verwendete Zeichensatz (ISO-8859-1) 

angegeben werden. Gemäss Syntaxregel in Kapitel 4 besteht das entsprechende Meta-

Attribut aus folgenden Elementen: 

Meta-Attribut-Element INTERLIS-Codefragment  

Zeilencode !!@ 

Meta-Attribut-Name ili.charset 

Meta-Attribut-Wert ISO-8859-1 

Ein vollständiges, gültiges Beispiel sieht also INTERLIS-Code wie folgt aus: 

INTERLIS 2.3; 

!!@ ili.charset = ISO-8859-1 

MODEL Example AT "mailto:noreply@localhost" VERSION "2010-02-30" = 

END Example. 

Das Meta-Attribut würde einem Objekt der Klasse «Model» aus dem Metamodell mit dem 

Namen «Example» zugeordnet: 

<IlisMeta07.ModelData BID="MODEL.Example"> 

  <IlisMeta07.ModelData.Model TID="Example"> 

    <Name>Example</Name> 

    <iliVersion>2.3</iliVersion> 

    <Contracted>false</Contracted> 

    <Kind>NormalM</Kind> 

    <At>mailto:noreply@localhost</At> 

    <Version>2010-02-30</Version> 

  </IlisMeta07.ModelData.Model> 

  <IlisMeta07.ModelData.MetaAttribute  

      TID="Example.METAOBJECT.ili.charset"> 

    <Name>ili.charset</Name> 

    <Value>ISO-8859-1</Value> 

    <MetaElement REF="Example" /> 

  </IlisMeta07.ModelData.MetaAttribute> 

</IlisMeta07.ModelData> 

7 Haftungsausschluss/Hinweise auf Rechte Dritter 

eCH-Standards, welche der Verein eCH dem Benutzer zur unentgeltlichen Nutzung zur Ver-

fügung stellt, oder welche eCH referenziert, haben nur den Status von Empfehlungen. Der 

Verein eCH haftet in keinem Fall für Entscheidungen oder Massnahmen, welche der Benut-

zer auf Grund dieser Dokumente trifft und / oder ergreift. Der Benutzer ist verpflichtet, die 

Dokumente vor deren Nutzung selbst zu überprüfen und sich gegebenenfalls beraten zu las-

sen. eCH-Standards können und sollen die technische, organisatorische oder juristische Be-

ratung im konkreten Einzelfall nicht ersetzen. 

In eCH-Standards referenzierte Dokumente, Verfahren, Methoden, Produkte und Standards 

sind unter Umständen markenrechtlich, urheberrechtlich oder patentrechtlich geschützt. Es 

liegt in der ausschliesslichen Verantwortlichkeit des Benutzers, sich die allenfalls erforderli-

chen Rechte bei den jeweils berechtigten Personen und/oder Organisationen zu beschaffen.  

Obwohl der Verein eCH all seine Sorgfalt darauf verwendet, die eCH-Standards sorgfältig 

auszuarbeiten, kann keine Zusicherung oder Garantie auf Aktualität, Vollständigkeit, Richtig-

keit bzw. Fehlerfreiheit der zur Verfügung gestellten Informationen und Dokumente gegeben 

werden. Der Inhalt von eCH-Standards kann jederzeit und ohne Ankündigung geändert wer-

den. 

Jede Haftung für Schäden, welche dem Benutzer aus dem Gebrauch der eCH-Standards 

entstehen ist, soweit gesetzlich zulässig, wegbedungen.  

 

8 Urheberrechte 

Wer eCH-Standards erarbeitet, behält das geistige Eigentum an diesen. Allerdings verpflich-

tet sich der Erarbeitende sein betreffendes geistiges Eigentum oder seine Rechte an geisti-

gem Eigentum anderer, sofern möglich, den jeweiligen Fachgruppen und dem Verein eCH 

kostenlos zur uneingeschränkten Nutzung und Weiterentwicklung im Rahmen des Vereins-

zweckes zur Verfügung zu stellen. 

Die von den Fachgruppen erarbeiteten Standards können unter Nennung der jeweiligen Ur-

heber von eCH unentgeltlich und uneingeschränkt genutzt, weiterverbreitet und weiterentwi-

ckelt werden.  

eCH-Standards sind vollständig dokumentiert und frei von lizenz- und/oder patentrechtlichen 

Einschränkungen. Die dazugehörige Dokumentation kann unentgeltlich bezogen werden. 

Diese Bestimmungen gelten ausschliesslich für die von eCH erarbeiteten Standards, nicht 

jedoch für Standards oder Produkte Dritter, auf welche in den eCH-Standards Bezug ge-

nommen wird. Die Standards enthalten die entsprechenden Hinweise auf die Rechte Dritter. 

